
EFX-TEK    www.efx-tek.com   teamefx@efx-tek.com 1 of 6

EZ-3 Timer Control Board
PN: 27558

Document Version: 1.0   01 MAY 2009 Copyright © 2009 EFX-TEK

DESCRIPTION

Prop builders have long used three-stage timers for simple props and effects.  The problem with traditional 
products is the inability to modify behavior beyond the original design.   The EZ-3 Timer Control Board solves this 
problem by converting the [re]programmable Prop-1, Prop-2, and Prop-SX controllers into simple or advanced 3-
stage timing control devices.

PRODUCT APPLICATIONS

 Props and holiday displays
 Special FX

FEATURES & BENEFITS

 Built-in potentiometer circuits to allow for trigger delay, on time, and off (reset) time
 Colored LEDs indicate timing stages
 Built-in manual switch for testing
 Terminal block provides for connection of external (N.O.) trigger device
 Pre-configured for Prop-1; may optionally be configured for the Prop-2 and Prop-SX
 Compact size: 1.5” x 2” (38.1mm x 50.8mm)

Mounting the EZ-3

Mount the EZ-3 by mating the 3x8 female socket with the 3x8 TTL header on the Prop-1, Prop-2, or Prop-SX.  
Secure the EZ-3 to the controller using a stand-off and screws (provided).

ULN2x03 Modification

The EZ-3 includes three potentiometer circuits that must be freed from the inputs of the ULN2x03 output driver 
used for valves and solenoids connected to the OUTx terminals.  Modification of the ULN2x03 is best handled 
with a small diagonal cutter.  Remove the pins as shown below.

ULN2803: Remove pins 1 through 5
ULN2003: Remove pins 1 through 4

Reconfiguring for the Prop-2/Prop-SX

The potentiometer circuits on the EZ-3 are pre-configured for the POT instruction used by the Prop-1 controller.  
To reconfigure for the RCTIME instruction used by the Prop-2 and Prop-SX you must bridge the “solder jumpers” 
with solder.  Using a low-wattage (25W) soldering iron and rosin-core solder bridge each of the solder jumpers; all 
six jumpers must be bridged for proper operation.

 CAUTION: Use caution when adding solder to the solder bridges on the EZ-3 as excessive heat can 
damage the PCB.  Use a low-wattage iron and just enough solder to make a connection between the 
inner pad and outer ring.   Using acid-core solder will damage the PCB.

Triggering

The EZ-3 – with the correct firmware in the prop controller – can be triggered thusly:

1. Press the MAN TR (manual trigger) button
2. Connect a N.O. button or switch (e.g., mat switch) to the EXT TR (external trigger) terminal block
3. Connect a N.O. or active-high device (e.g., PIR sensor) to the TRIG header (P6)

EFX-TEK recommends that the input on P6 be debounced to prevent false triggering on a spurious input.  See 
the demonstration program at the end of this document for an example of input debouncing.

MECHANICAL SPECIFICATIONS 

EZ-3 Schematic

Accessories

805-00035 14” Extension Cable
805-00175 68” Extension Cable

Additional Applications

For additional ideas and application notes for the EZ-3 Timer Control Board be sure to visit us on the Internet at 
the following links:

    www.efx-tek.com
    forums.efx-tek.com

Errata



EFX-TEK    www.efx-tek.com   teamefx@efx-tek.com 4 of 6

Demonstration Program

The following Prop-1 program allows the EZ-3 to control a standard, 3-stage timer application.  The trigger delay 
time may be set from zero to 10 seconds, the on time from 1 to 60 seconds, and the off time from 1 to 20
seconds.

 NOTE: If you’d like custom timing for a 3-stage application using the EZ-3 simply post a programming 
request with details in our Tech Support forums.

' =========================================================================
'
'   File...... EZ-3_Standard.BS1
'   Purpose...
'   Author.... EFX-TEK
'              Copyright (c) 2009 EFX-TEK
'              Some Rights Reserved
'              -- see http://creativecommons.org/licenses/by/3.0/
'   E-mail.... teamefx@efx-tek.com
'   Started...
'   Updated... 01 MAR 2009
'
'   {$STAMP BS1}
'   {$PBASIC 1.0}
'
' =========================================================================

' -----[ Program Description ]---------------------------------------------
'
' Standard 3-stage timer application.
'
' Trigger Delay... 0 TO 10 seconds
' On Time......... 1 TO 60 seconds
' Off Timer....... 1 TO 20 seconds
'
' For best accuracy of each stage you should run the POT Scaling program
' from the Run menu for each stage.  Set the scale value into the POT
' instruction for each stage.
'
' Trigger Delay... Pin 5
' On Time......... Pin 4
' Off Timer....... Pin 3
'
' Turn each pot fully clockwise and use the scale value provided.  The
' scale value will be around 100 but will vary from pot-to-pot based on
' minor component differences.

' -----[ Revision History ]------------------------------------------------

' -----[ I/O Definitions ]-------------------------------------------------

SYMBOL  Sio             = 7                     ' SETUP = out; no ULN
SYMBOL  Trigger         = PIN6                  ' SETUP = DN
SYMBOL  AdjTrDelay      = 5                     ' on delay  (no ULN)
SYMBOL  AdjOnTime       = 4                     ' run time  (no ULN)
SYMBOL  AdjOffTime      = 3                     ' off delay (no ULN)
SYMBOL  TrDelay         = PIN2                  ' trigger delay output
SYMBOL  OnTime          = PIN1                  ' on time output
SYMBOL  OffTime         = PIN0                  ' off time output

' -----[ Constants ]-------------------------------------------------------

SYMBOL  IsOn            = 1                     ' for active-high in/out
SYMBOL  IsOff           = 0

SYMBOL  Yes             = 1
SYMBOL  No              = 0

' -----[ Variables ]-------------------------------------------------------

SYMBOL  timer           = W0
SYMBOL   negFlag        =  BIT15

SYMBOL  tix             = W1

' -----[ Initialization ]--------------------------------------------------

Reset:
  PINS = %00000000                              ' clear all
  DIRS = %00000111                              ' set outputs

' -----[ Program Code ]----------------------------------------------------

Main:
  timer = 0                                     ' reset timer

Check_Trigger:                                  ' debounce input
  PAUSE 5                                       ' loop pad
  timer = timer + 5 * Trigger                   ' update timer
  IF timer < 100 THEN Check_Trigger             ' wait for 0.1 sec input

' Trigger delay is 0 to 10 seconds
' -- convert pot span (0 to 250) to 0 to 100 "tix"
' -- tix = timer x 0.4

Trigger_Delay:
  POT AdjTrDelay, 100, timer                    ' raw input, 5 to 255
  timer = timer - 5                             ' correct low end offset
  IF negFlag = No THEN Run_TD
    timer = 0                                   ' fix roll-under

Run_TD:
  IF timer = 0 THEN On_Time                     ' skip if zero
  tix = timer ** $6666                          ' tix = timer x 0.4
  TrDelay = IsOn
  GOSUB Delay_Tix
  TrDelay = IsOff

' On time is 1 to 60 seconds
' -- convert pot span (0 to 250) to 10 to 600 "tix"
' -- tix = (timer x 2.36) + 10

On_Time:
  timer = 0
  POT AdjOnTime, 100, timer
  timer = timer - 5
  IF negFlag = No THEN Run_OT
    timer = 0

Run_OT:
  tix = timer * 2                               ' tix = timer x 2

  timer = timer ** $5C28                        ' x 0.36
  tix = tix + timer                             ' tix = timer x 2.36
  tix = tix + 10                                ' set low end
  OnTime = IsOn
  GOSUB Delay_Tix
  OnTime = IsOff

' Off time is 1 to 20 seconds
' -- convert pot span (0 to 250) to 10 to 200 "tix"
' -- tix = (timer x 0.76) + 10

Off_Time:
  timer = 0
  POT AdjOffTime, 100, timer
  timer = timer - 5
  IF negFlag = No THEN Run_XT
    timer = 0

Run_XT:
  tix = timer ** $C28F                          ' tix = timer x 0.75
  tix = tix + 10                                ' set low end
  OffTime = IsOn
  GOSUB Delay_Tix
  OffTime = IsOff

  GOTO Main

' -----[ Subroutines ]-----------------------------------------------------

' Delay in 0.1-second units
' -- put delay in "tix" before calling

Delay_Tix:
  IF tix = 0 THEN DT_Exit
    PAUSE 100
    tix = tix - 1
    GOTO Delay_Tix

DT_Exit:
  RETURN


